DECUS

PROGRAM LIBRARY

DECUS NO.	8-590
TITLE	MATRIX INVERSION
AUTHOR	Jean-Francois Echallier and Jacques Pernier
COMPANY	I.N.S.E.R.M. Bron, France
DATE	November 17, 1972
SOURCE LANGUAGE	PAL

MATRIX INVERSION

ABSTRACT.

This subroutine inverts a matrix of real numbers residing in core in floating format (three words). The Gauss-Jordan method of elimination is used to perform the inversion.

REQUIREMENTS

Storage.

This subroutine occupies 737_8 + 2.NØ locations anywhere in core and four locations in page 0. NØ is the matrix order. The beginning of the subroutine must be at a page boundary. The listing which accompanies this write-up shows the subprogram beginning at location 200_8. This subroutine uses the basic Floating Point Package (F.P.P.) and a patch of 7 instructions (RFSGE) which correspond to an additional operation in F.P.P. and must be in the same field that the F.P.P.

If F.P.P. is not in the same memory field than the routine one, the extended memory patch (DEC 8 364) must be used.

Equipment : PDP8 I or E with EAE or
PDP8 L with instruction set simulator (DEC 8 - 17U).

USAGE OF THE SUBROUTINE.

Loading.

The PAL symbolic tape concerns the subroutine only, and may be assembled into an alternate location. (Warning : don't forget to keep free 2NØ locations after the end of the subroutine).

Assembling parameters.

Before the assembling, let set up the following parameters :

FPPFLD = X where X is equal to ØØ, 10, 20, ... etc according to the F.P.P. resident field Ø, 1, 2 etc....

CURFLD = Y where Y is equal to ØØ, 1Ø, 2Ø,.... etc according to the subroutine, the calling program and data memory field (the same for the 3).

Calling sequence.

The elements a_{ij} of the matrix must be in order a_{11}, a_{12},... a_{21}, a_{22}.... a_{n1},.... a_{nn}. (i.e. row after row).

Initialize :

```
ADRØ,...      / first address of the matrix array
NØ,..         / matrix order
JMS (INVMAT)
```

Return will be to PC+1 with the inverse of matrix in place of the original matrix array and the determinant is returned as the value of DET.

Utilization of internal subroutines.

Some internal subroutines may be used for deposing data into the matrix array, or picking up results from the same array.

Warning : The first element is a_{oo} and the last $a_{n-1, n-1}$. (instead of a_{11} to $a_{n,n}$ usually used).

- Deposing data : Initialize Ii and J
  JMS (RCALij)

Return will be to PC+1 with the address of the element (a_{ij}) in PTAR (the address of the exponent).

In the same way, we can use for an element $a_{k,j}$ the routine RCAL KJ after initializing K and J. The address of the element will be in PTAIJ.

A third subroutine RCALIK is disponible for an element $a_{i,k}$. The address is in PTAIJ.

- Picking up data : As for deposing data.

- Utilization of loops tests : If INVMAT has not still been used, initialize N = NØ - 1.

Test : J = N ?

PC	JMS	R TEST J / J = N ?
PC + 1	RETURN FROM TEST / J $\neq$ N	
PC + 2	RETURN FROM TEST / J = N	

Test : II = N ?

PC	JMS	R TEST I / II = N ?
PC + 1	RETURN FROM TEST / II $\neq$ N	
PC + 2	RETURN FROM TEST / II $\neq$ N	

RESTRICTIONS.

The calling program, the subroutine and the data must be in the same field.

If we use an other version than YQYA-PB for the F.P.P., let's verify that the entry point (for additional pseudo-operator FSGE = 11 in F.P.P.) of RFSGE may be in address 6555 of TABLE 6 of the F.P.P. One can change the code of FSGE and the address entry point in TABLE 6 of the F.P.P.

USAGE OF THE PROGRAM (BINARY PAPER TAPE).

Loading.

A routine which allows input of data and output of results on T.T.Y. is supplied as a binary tape which may be loaded with the BIN loader after the F.P.P.

- Operating procedure.

. Starting address : 1ØØØ

. Respond to question

MATRIX ORDER :

. Type values of coefficients row after row. Each value will be separated by a space.

. Respond to question.

CHANGE VALUE ? (Ø = No ; 1 = yes)

. The coefficient of the inverse matrix are printed out row after row.

. Respond to question : INVERSE INVERSE ? (Ø = No ; 1 = yes).

- Informations.

. Time of execution (see figure).

. Example of utilization with a matrix of which value of determinant is near 0.

We use the pseudo F.P.P. operators INPUT = 13 and OUTPUT = 14, and the F.P.P. used must be accomoded for (in TABLE6).



EXECUTION TIME

```
MATRIX ORDER:4

   3     8     5     6

   3     8     5     5.995

   4     6     8     8

   4     2     9     7

CHANGE VALUE?0

INVERSE MATRIX

   +0.3020882E+04   -0.3001876E+04   -0.3101118E+02   +0.1700648E+02

   -0.4022507E+03   +0.4002499E+03   +0.3501489E+01   -0.2000865E+01

   -0.1408877E+04   +0.1400874E+04   +0.1300521E+02   -0.7003027E+01

   +0.2001247E+03   -0.2001242E+03   -0.7400512E-03   +0.4293918E-03

INVERSE INVERSE?1

   +0.2998024E+01   +0.7994611E+01   +0.4996741E+01   +0.5996038E+01

   +0.2998026E+01   +0.7994611E+01   +0.4996742E+01   +0.5991041E+01

   +0.3996766E+01   +0.5991228E+01   +0.7994650E+01   +0.7993520E+01

   +0.3996387E+01   +0.1990233E+01   +0.8994015E+01   +0.6992763E+01

MATRIX ORDER:4

   3     5     2     1

   6     8     3     12

   45     7     9     4

   5     3     9     1

CHANGE VALUE?1
ROW: 3
COLUMN:1
VALUE:8

CHANGE VALUE?0

INVERSE MATRIX

   -0.5833323E+00   -0.2247471E+00   +0.1025250E+01   -0.8207057E+00

   +0.4999994E+00   +0.7575742E-01   -0.4242416E+00   +0.2878781E+00

   +0.1666664E+00   +0.8585850E-01   -0.4141409E+00   +0.4595955E+00

   -0.8333335E-01   +0.1237373E+00   -0.1262625E+00   +0.1035352E+00

INVERSE INVERSE?0

MATRIX ORDER:
```

```
.R PIP

TTY:<INVMAT
```

```
/ROUTINE D'INVERSION DE MATRICE
FPPFLD=00 /RESIDENT FIELD OF THE F.P.P.
CURFLD=00 /FIELD WHERE ARE THE MATRIX & THE MAT. INV. ROUTINE
/ADR0 , CONTAIN  THE 1ST ADRESSE OF THE MATRIX ;
/N0 ,   THE ORDER OF THE MATRIX
/
/THE ROUTINE USES A BUFFER EQUAL TO 2*N0 JUST AFTER ITS LAST
/ ADRESSE OR AFTER THE I/O SUBPROGRAMME LAST ONE ,IF IN CORE
/CALLING SEQUENCE
/JMS  I (INVMAT)
/  RETURN FROM THE ROUTINE  WITH THE INVERTED MATRIX
/           AT THE SAME PLACE   OR WITH DET=0 IF
/           THE DETERMINANT IS NUL
/DATA FIELD IS RESET TO (CALFLD)  ; AC=0
/
FIELD 0  /SET TO 0,1,2.. ACCORDING TO FPP RESIDENT FIELD
*7
5600
*6555
RFSGE
/
/
*4400  /EVERY FREE PLACE IS GOOD
RFSGE,HLT /REALIZE A SKIP IF FAC > OR = TO 0
  CLA
  TAD 45
  SMA CLA
  ISZ I A5655
  JMP I RFSGE
A5655,5655
/
FSGE=11
/
FIELD 0 / SET TO 0,1,2,.. ACCORDING TO (CURFLD)
*70   /CONSTANTS IN ZERO PAGE NOT NECESSARLY AT (70)
PTAIJ,0
PTAR,0
PTJK,0
PTIK,0
*200
INVMAT,HLT
  CIF+FPPFLD
  JMS I 7
  FGET FUN
```

```
  FPUT DET /DETERMINANT=1.
  FEXT
  TAD PTIK0
  TAD N0
  DCA PTJK0
  CLA CMA
  TAD N0
  DCA N
  JMS INVER
  JMP DETNUL /RETOUR SI DETERMINANT=0
  DCA L  /RESTAURER L'ORDRE DE LA MATRICE
RESTAU, TAD L
  CMA IAC / K=N-L+1
  TAD  N /  ON DOIT METTRE N-L CAR ON PART  A L=0
  DCA K
  TAD PTIK0
  TAD K
  DCA PTIK
  TAD I PTIK
  DCA J  /J=IK(K)
  JMS RJMK  /J-K
  SPA SNA CLA
  JMP .+7
  DCA II  /I=0
  JMS RCALIK
  JMS RCALIJ
  JMS RSAVAA /HOLD=A(I,K) ;A(I,K)=A(I,J) ;A(I,J)=-HOLD
  JMS RTESTI  /I=N ?
  JMP .-4 /NON  I=I+1 FAIT DANS TESTI
  TAD PTJK0
  TAD K
  DCA PTJK
  TAD I PTJK
  DCA II  /I=JK(K)
  JMS RIMK  /I-K
  SPA SNA CLA
  JMP .+7
  DCA J  //J=0
  JMS RCALKJ
  JMS RCALIJ
  JMS RSAVAA  /HOLD=A(K,J) ;A(K,J)=-A(I,J) ;A(I,J)=SAVE
  JMS RTESTJ  /J=N ?
  JMP .-4 /NON
  TAD K  /SI K=0  : L=N
  SNA CLA /L=N ?
  JMP I INVMAT  /OUI EXIT FIN DE L'INVERSION
  ISZ L
  JMP RESTAU  /NON
/
DETNUL,DCA DET
  DCA DET+1
  DCA DET+2
  JMP I INVMAT

RIMK,HLT /CACUL DE I-K
  TAD K
  CIA
  TAD II
  JMP I RIMK /RESULT DANS AC
```

```
/
RJMK,HLT /CALCUL DE J-K
  TAD K
  CIA
  TAD J
  JMP I RJMK
/

RCALIJ,HLT /CALCUL DE L'ADRESSE DE ( A(I,J) )
  TAD II
  JMS RMULTN   /I*N
  TAD J

  JMS MUY3P / (I (I*N)+J)*3 +ADRESSE DEPART
  DCA PTAR
  JMP I RCALIJ
/
RCALIK,HLT /CALCUL DE L'ADRESSE DE ( A(I,K) )
  TAD II
  JMS RMULTN
  TAD K
  JMS MUY3P
  DCA PTAIJ
  JMP I RCALIK
/
RCALKJ,HLT  / POUR ELEMENT ( A(K,J) )
  TAD K
  JMS RMULTN
  TAD J
  JMS MUY3P
  DCA PTAIJ
  JMP I RCALKJ
/
MUY3P,HLT/ NB PRIS DANS AC *3 +ADRESSE DEPART(ADRØ)
  MQL MUY
  3
  MQA
  TAD ADRØ
  JMP I MUY3P /RETOUR AVEC RESULTAT DANS AC
/
RTESTN,HLT
  TAD I RTESTN /RECUPERER L'ARGUMENT SUIVANT LE JMS
  CMA IAC
  TAD N
  SNA CLA  /ARG=N ?
  JMP .+3  /OUI
  ISZ I RTESTN/INCREMENTER L'ARGUMENT
  ISZ RTESTN /+1ADRESSE RETOUR
  ISZ RTESTN /    "
  JMP I RTESTN/RETOUR A ARG+1 SI ARG=N; ARG+2 SI ARG DIF. DE N
```

```
/
/
/
RMULTN,HLT
  MQL MUY
NØ,3
  MQA
  JMP I RMULTN
/
ADRØ,1000
N,2
L,Ø
PTIKØ,DEBIK
PTJKØ,Ø
DET,Ø;Ø;Ø
/
PAGE
/
/
/ROUTINE D'INVERSION PPT DITE
INVER,HLT
  DCA K
A1,DCA AMAX+1
  DCA AMAX+2
  TAD K  /RECHERCHE DE L'ELEMENT MAX DE CHAQUE LIGNE ONREPERE
  DCA II / LIGNE ET COL DE L'ELEMENT MAX DANS IK(K) ET JK(K)
LIGSUI,TAD K
  DCA J  /J=K
COLSUI,JMS RCALIJ
  TAD PTIKØ
  TAD K
  DCA PTIK  /POINTEUR SUR  IK(K)
  TAD PTJKØ
  TAD K
  DCA PTJK  /  "  "  JK(K)
  JMS ABSVL / CALCUL  VAL. ABS. DE AMAX ET A(I,J)
  JMS RTESTJ /J=N ?
  JMP COLSUI /NON
  JMS RTESTI /I=N ?
  JMP LIGSUI /NON
  CIF+FPPFLD
  JMS I 7
  FGET AMAX
  FEXT
  TAD 45
  SNA CLA
  JMP I INVER /OUI EXIT
  TAD I PTIK
  DCA II  /I=IK(K)
  JMS RIMK  /I-K
  SPA SNA CLA
  JMP JJKK  /I-K =Ø
  DCA J    /I-K >Ø
  JMS RCALKJ
  JMS RCALIJ
  JMS RHOLAA /HOLD=A(K,J) ;A(K,J)=A(I,J) ;A(I,J)=-HOLD
  JMS RTESTJ /J=N ?
  JMP .-4
```

```
JJKK,TAD I  PTJK
  DCA J  /J=JK(K)
  JMS RJMK
  SPA SNA CLA
  JMP ASSURM
  DCA II
  JMS RCALIK
  JMS RCALIJ
  JMS RHOLAA
  JMS RTESTI  /I=N ?
  JMP .-4
ASSURM,DCA II /I=0
  JMS RIMK
  SNA CLA  /I-K =0 ?
  JMP .+11
  JMS RCALIK
  CIF+FPPFLD
  JMS I 7
  FGET I PTAIJ
  FMPY FM1
  FDIV AMAX
  FPUT I PTAIJ  /A(I,K) =-A(I,K)/AMAX
  FEXT
  JMS RTESTI /I=N ?
  JMP ASSURM+1
APASA,DCA II
  DCA J
  JMS RIMK /ATTENTION IL Y A 1 JMP APASA+1 ET +2
  SNA CLA
  JMP FINBOU
  JMS RJMK
  SZA CLA  /J-K=0 ?
  JMS APAMA /NON
FINBOU,JMS RTESTJ
  JMP APASA+2
  JMS RTESTI
  JMP APASA+1
  DCA J
DIVAMA,JMS RJMK
  SNA CLA
  JMP .+10
  JMS RCALKJ
  CIF+FPPFLD
  JMS I 7
  FGET I PTAIJ
  FDIV AMAX
  FPUT I PTAIJ
  FEXT
  JMS RTESTJ
  JMP DIVAMA
   TAD K
  JMS RMULTN
  TAD K
  JMS MUY3P
  DCA PTAR
  JMS DETAMA
  JMS RTESTN
  K,0
  SKP
  JMP A1
```

```
  ISZ INVER
  JMP I INVER
/

/
RTESTJ,HLT
  JMS RTESTN
  J,0
  ISZ RTESTJ
  JMP I RTESTJ
/
RTESTI,HLT
  JMS RTESTN
  II,0
  ISZ RTESTI
  JMP I RTESTI

/
AMAX,0;0;0
/
PAGE
/
/ROUTINES
RSAVAA,HLT  /CALCULE : HOLD= (I PTAIJ) ;(I PTAIJ )= -(I PTAR)
  CIF+FPPFLD
  JMS I 7   /         (I PTAR) = HOLD
  FGET I PTAIJ
  FPUT HOLD
  FGET I PTAR
  FMPY FM1
  FPUT I PTAIJ
  FGET HOLD
  FPUT I PTAR
  FEXT
  JMP I RSAVAA
/
/CALCUL DE HOLD= (I PTAIJ)
/          (I PTAIJ) =(I PTAR)  ;(I PTAR) =-HOLD
RHOLAA,HLT
  CIF+FPPFLD
  JMS I 7
  FGET I PTAIJ
  FPUT HOLD
  FGET I PTAR
  FPUT I PTAIJ
  FGET  HOLD
  FMPY FM1
  FPUT I PTAR
  FEXT
  JMP I RHOLAA
/


ABSVL,HLT
  CIF+FPPFLD
  JMS I 7
  FGET I DAMAX
  FSGE  /FAC>=0 ?
```

```
    FMPY FM1 /NON
    FPUT FTEMP1
    FGET I PTAR
    FSGE
    FMPY FM1
    FSUB FTEMP1
    FEXT
    TAD 45
    SPA CLA /FABS(MAX) < OU = FABS (A(I,J)) ?
    JMP I ABSVL /NON
    CIF+FPPFLD
    JMS I 7
    FGET I PTAR /OUI  ,PRENDRE A(I,J) COMME MAX
    FPUT I DAMAX
    FEXT
    TAD I DII
    DCA I PTIK /RELEVER LES INDICES
    TAD I DJ
    DCA I PTJK
    JMP I ABSVL
/
/


/
DETAMA,HLT
    CIF+FPPFLD
    JMS I 7
    FGET FUN
    FDIV I DAMAX
    FPUT I PTAR / 1./AMAX
    FGET I DDET
    FMPY I DAMAX
    FPUT I DDET
    FEXT
    JMP I DETAMA
/
/
APAMA,HLT
    JMS I CALIK
    TAD PTAIJ
    DCA PTAR
    JMS I CALKJ
    CIF+FPPFLD
    JMS I 7
    FGET I PTAIJ
    FMPY I PTAR
    FEXT
    JMS I CALIJ
    CIF+FPPFLD
    JMS I 7
    FADD I PTAR
    FPUT I PTAR /A(I,J)=A(I,J)+A(I,K)*A(K,J)
FEXT
    JMP I APAMA
```

```
/
HOLD,0;0;0
FM1,1;6000;0
FUN,1;2000;0
FTEMP1,0;0;0
CALIK,RCALIK
CALIJ,RCALIJ
CALKJ,RCALKJ
DAMAX,AMAX
DDET,DET
DII,II
DJ,J
/
DEBIK=.
/
PAUSE
/
```